Let's Make a Deal!

Imagine that you are transported back in time and, as the mist clears, you see that you are on stage, the finalist, on the game show *Let's Make a Deal*. Monty makes small talk and tries to calm you then gestures toward doors "*One*, *Two*, and *Three*," explaining that behind one of them is the day's grand prize, the *Big Deal* (customarily, a trip to Hawaii or a new, well a 1972, luxury automobile) and behind the others (he need not explain, you know from having watched too much daytime television that they conceal a one year supply of denture cream or laxatives). You may choose one of the three doors and whatever lies behind is yours! So, you carefully consider everything your grandmother told you to do if ever in this situation, then choose your door. Monty is impressed and, also knowing which door conceals the real prize and wanting to add drama, has one of the doors you did not choose opened. He then offers to exchange the remaining door for yours. Should you swap? Do your odds of winning the grand prize change by exchanging doors?

Thoughts:

Let C symbolize the door you chose, N_1 and N_2 the doors you did not choose. The events "C gets prize" and "N_1 or N_2 gets prize" are disjoint since only one door has the grand prize. Assuming your choice was unbiased, P(C gets prize)=1/3 and P(N_1 or N_2 gets prize)=2/3. You have C (and a 1/3 chance of winning) and Monty has N_1 and N_2 (and a 2/3 chance of having the grand prize door). Monty knows which door has the real prize and, if in possession of it, will *always* make it available to you (by always opening a door with a *zonker* and never revealing the grand prize). There is a 2/3 chance that he has the prize winning door and a 100% chance that he will offer it to you whenever he has it. This gives the door that he offers a 2/3 chance of winning, double the odds of the one you possess.

Good in theory. Let's test it.

Simulation outline:

Generate two columns of random integers (values 1, 2, or 3), one column (D) to represent the actual winning door, the other to represent your (random, unbiased) choice (C). Let i be a row index into D and C. Then, pairing i^{th} row entries, there are three possible pairs with equal entries (1,1; 2,2; and 3,3) out of 3X3=9 total possible pairs. These three symbolize your chosen door being the same as the winning door. So, these simulated columns agree with our expectation that your choice has a 1/3 chance of being correct. Now generate a third column, Q, to represent the door from those not chosen by you (not in column C) and not revealed by Monty. In its i^{th} row put one of the following:

1. D_i when $C_i <> D_i$ since Monty will never reveal the winning door.

   or

2. A random (50% equal probability) selection of the remaining two doors when $C_i = D_i$. This simulates Monty arbitrarily choosing from the two doors he possesses (knowing neither contain the grand prize).

Finally, calculate the ratio of pairs D_i, Q_i with equal entries in all rows. This is the probability of your winning the grand prize by swapping doors.

Results:

Following is a SAS program to simulate 1,000 *Big Deals*:

```
/* Let's Make a Deal!
   Simulate 1,000 Big Deals
   Generate random door choices and remainders in separate steps to avoid any possible
   correlation */

/* First, column C, your choice of doors */
data doors(keep=C);
  call streaminit(1234);
  do i=1 to 1000;
    C=floor(rand('uniform')*2.999999)+1;
   output;
  end;
run;

/* Now, column D, the actual winning door */
data doors(keep=C D);
  call streaminit(4321);
  set doors;
  D=floor(rand('uniform')*2.999999)+1;
run;

/* And now column Q, the one remaining after Monty reveals one of the zonkers */
data doors(keep=C D Q r);
  call streaminit(7890);
  set doors;
  if(D~=C)then
    /* You did not choose the winning door, so Monty makes the winning door available for
       swap by revealing his non-winner */
    Q=D;
  else
    /* You did choose the winning door, so Monty arbitrarily chooses one of his (non-
       winners) to be available for swap */
    do;
     r=rand('uniform');
     if(C=1)then if(r<.5)then Q=2; else Q=3;
     else if(C=2)then if(r<.5)then Q=1; else Q=3;
     else if(C=3)then if(r<.5)then Q=1; else Q=2;
   end;
run;

/* Verify random door generation
   Columns C and D should contain equal proportions (1/3) of ach door */
proc sql;
  /* Report proportion of each door in chosen and actual columns */
  select D as WinningDoor, count(1) from doors group by D;
  select C as ChosenDoor, count(1) from doors group by C;
  /* Report proportion of chosen doors that are winners */
  select count(1)/(select count(1) from doors) as ProportionWinningDoorsChosen
  from   doors
  where  D=C;
  /* Report proportion of each door not revealed by Monty */
  select Q as MontyRetainedDoor, count(1) from doors group by Q;
quit;

/* Now report the proportion of rows where Qi=Di.
   This is the probability of winning after swapping doors, since this is the proportion
   of events (outcomes in such contests) where the contestant would receive the winning
   door.  In trials, this reports a consistent 2/3 probability of receiving the winning door
   after swapping. */
proc sql;
  select count(1)/(select count(1) from doors) from doors where Q=D;
quit;

/* More detail */
proc univariate data=doors;
  histogram D C Q;
run;

/* Let's chart the resulting proportions */
data swap(keep=outcome);
  set doors;
```

```
  if(Q=D)then outcome='Swap=Win '; else outcome='Swap=Lose';
run;

title "Let's Make a Deal!";
title2 'Simulated Outcome of Swapping Doors - 1,000 Contests';
pattern1 color=white;
pattern2 color=white;

proc gchart data=swap;
  pie outcome / noheading value=inside type=pct;
run;
quit;
```

Here's a sample of the randomly generated doors table:

C (your choice)	D (winning door)	Q (available for swap)
3	3	2
3	2	2
1	3	3
1	1	3
3	1	1
1	3	3
1	2	2
2	1	1
1	2	2
2	2	1

Notice that, even in this small sample of 10 contests, the chosen door is the winning one three times, while the door available for exchange is the winner seven times. Exchanging doors doubles the odds of winning.

Finally, here is a pie chart indicating the proportion of simulated contests where exchanging doors with Monty causes a win or a loss (I hope grandma's advice was to swap doors!).